

Você está aqui: [Configuração e manutenção](#) > [Criando aplicativos, questionários e soluções](#) > [Campos](#) > Funções do sistema

Funções do sistema

As seguintes funções do sistema funcionam com variáveis específicas do Archer e opções para produzir resultados dinâmicos.

Nesta página

- [Função COMBINESELECTIONS](#)
- [Função CONTAINS](#)
- [Função CONTENTID](#)
- [Função GETGROUPS](#)
- [Função GETUSERS](#)
- [Função GROUP](#)
- [Função ISCORRECT](#)
- [Função ISEMPY](#)
- [Função ISNUMBER](#)
- [Função MOSTRECENTVALUE](#)
- [Função NOVALUE](#)
- [Função OTHERTEXT](#)
- [Função REF](#)
- [Função SELECTEDVALUENUMBER](#)
- [Função TRACKINGID](#)
- [Função USER](#)
- [Função USERDEFAULTEMAIL](#)
- [Função USERFIRSTNAME](#)
- [Função USERLASTNAME](#)
- [Função USERMIDDLENAME](#)
- [Função VALUEOF](#)
- [Função WEIGHTEDSCORE](#)
- [Função WEIGHTING](#)

Função COMBINESELECTIONS

A função COMBINESELECTIONS mescla os valores selecionados de todos os campos de lista de valores incluídos na função para 1 ou mais aplicativos.

Importante: O campo Lista de valores calculado deve conter todos os valores da Lista de valores das Listas de valores referenciadas para que todos os Valores selecionados sejam exibidos. Além disso, a função executa uma comparação entre strings que diferencia maiúsculas de minúsculas nos Valores selecionados; portanto, é possível que os campos Lista de valores referenciados na função consultem Listas de valores diferentes.

Tipo de retorno: Lista de valores

Sintaxe: COMBINESELECTIONS([campolista_valores1], [campolista_valores2],[...])

Na sintaxe acima, a expressão deve conter pelo menos 1 campo de lista de valores.

A tabela a seguir descreve os parâmetros da função COMBINESELECTIONS.

Parâmetro	Descrição
campolista_valores1	Os campos da lista de valores que estão sendo combinados.
campolista_valores2	Listas de valores adicionais, até 255 itens no máximo. Os campos de lista de valores devem ser separados por vírgulas.

Exemplo:

A tabela a seguir fornece um exemplo de fórmula da função COMBINESELECTIONS.

Fórmula	Resultado
COMBINESELECTIONS(REF([AppB],[Cores])) em que a lista de valores de cores contém Vermelho e Verde em 1 registro relacionado e Verde e Azul em outro registro relacionado do aplicativo com referência cruzada.	Vermelho Verde Azul

Função CONTAINS

A função CONTAINS é usada para determinar se algum valor de uma lista de valores corresponde ao valor armazenado em um campo fornecido. Se 1 dos valores corresponder ao valor do campo, a função será avaliada como TRUE e 1 valor será retornado. Se nenhum valor da lista corresponder ao do campo, a função será avaliada como FALSE e outro valor será retornado. A função CONTAINS deve ser usada junto com a função IF.

Embora não seja frequente, a função CONTAINS pode ser usada para testar se o valor da string de um campo Texto fornecido é igual a algum valor de 1 lista de valores de string fornecidos. Quando o destino é um campo Texto o valor da string no campo será comparado às strings de teste fornecidas. A comparação será baseada apenas em strings completas. A função não determinará correspondências baseadas em uma substring. Por exemplo, se o campo de Texto de destino contiver o valor "Laptop confiscado", a função CONTAINS não identificará um valor correspondente a uma string de teste "laptop".

Tipo de retorno: TRUE ou FALSE

Sintaxe: IF(CONTAINS(**tipo_avaliação**, **ref_campo**, **valor1**, **valor2..**),**valor_se_true**,**valor_se_false**)

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve os parâmetros da função CONTAINS.

Parâmetro	Descrição
tipo_aval	<p>Uma destas palavras-chave de tipo de avaliação:</p> <ul style="list-style-type: none">• ANY. Especifica que o campo de destino deve conter pelo menos 1 dos valores selecionados.• EXACT. Especifica que a função de destino deve conter todos os valores selecionados e somente esses valores• ALL. Especifica que o campo de destino deve conter no mínimo, todos os valores selecionados. Com a palavra-chave ALL, a função CONTAINS retornará TRUE mesmo existindo valores selecionados no campo além daqueles especificados. <p>Observação: Na avaliação de um campo Texto, a função CONTAINS retornará TRUE se 1 ou mais strings de teste</p>

Parâmetro	Descrição
	<p>fornecidas corresponderem ao valor de string do campo Texto. As palavras-chave EXACT e ALL têm o mesmo comportamento na avaliação de um campo de texto. Quando elas são usadas para avaliar um campo Texto, apenas 1 string de teste deve ser especificada porque campos Texto somente podem ter 1 valor.</p>
ref_campo	<p>Referência a um campo, por exemplo, [nome campo]. A referência deve ser Lista de valores, Lista de usuários/grupos, Permissões de registro ou Campo de texto.</p>
valor1, valor2, . . .	<p>Um dos seguintes:</p> <ul style="list-style-type: none"> • Um ou mais valores de string fornecidos como possivelmente correspondentes aos valores selecionados no campo da lista de destino. (Na avaliação dos valores selecionados em um campo do tipo Lista de valores, os valores de string de teste devem ser incluídos na função VALUEOF.) • Uma ou mais referências aos campos Lista de valores, Lista de usuários/grupos ou Permissões de registro. Em tempo de execução, o sistema extrairá a lista dos valores selecionados nos campos fornecidos e considerará os valores como strings (ou IDs de usuário/grupo) a serem testadas e comparadas aos valores selecionados no campo de destino. • Observação: Quando se faz referência a vários campos para obter valores de teste, todos os campos devem ser do mesmo tipo e iguais aos do campo de destino. Quando se faz referência a um campo Lista de usuários/grupos ou Permissões de registro para obter valores de teste, a referência ao campo deve ser retornada na função GETUSERS ou GETGROUPS para fornecer o contexto adequado para recuperação dos valores selecionados. Os valores selecionados em campos Lista de usuários/grupos e Permissões de registro serão retornados como IDs, e os selecionados em um campo Lista de valores serão retornados como strings.

Parâmetro	Descrição
	<ul style="list-style-type: none"> Combinação de referências a valores literais e de campo. Em tempo de execução, o sistema extrairá a lista de valores selecionados de qualquer campo referido e, conceitualmente, combinará os valores selecionados com os valores literais fornecidos para formar uma lista única de valores selecionados para testar o campo de destino.

Exemplos:

A tabela a seguir fornece exemplos de fórmulas da função CONTAINS.

Fórmula	Resultado
IF(CONTAINS(ANY, [Local], VALUEOF([Escritório], "Chicago", "Local", "Global"))) cujo valor selecionado em Local é Chicago.	Local
IF(CONTAINS(ANY, [Região], VALUEOF([Escritório], "Connecticut", "New York", "Massachusetts"), VALUEOF([Escritório], "Kansas", "Illinois", "Texas")))	Connecticut New York Massachusetts

Função CONTENTID

A função CONTENTID retorna o ID de registro de um conteúdo, que identifica exclusivamente o registro atual dentro do contexto do aplicativo ou subformulário atual. IDs de conteúdo geralmente são sequenciais, começando do número 1.

Tipo de retorno: Numérico

Sintaxe: CONTENTID()

Esta função não tem nenhum parâmetro.

Exemplo:

A tabela a seguir fornece um exemplo de fórmula da função CONTENTID.

Fórmula	Resultado
[Código do tipo] & "-" & TRACKINGID() & "-" & CONTENTID() cujo valor no campo Código do tipo é WORM, o ID de controle de todo o sistema é 678904 e o ID de controle específico do aplicativo é 34.	WORM-678904-34

Função GETGROUPS

A função GETGROUPS retorna uma lista de IDs de grupo dos grupos atualmente selecionados em um campo especificado, do tipo Lista de usuários/grupos ou Permissões de registro. A lista de IDs de grupo pode ser, assim, avaliado por outra função. Por exemplo, a função GETGROUPS pode ser usada dentro de uma função CONTAINS para avaliar se um determinado grupo está contido na lista de grupos recuperada de um campo especificado, do tipo Lista de usuários/grupos.

Tipo de retorno: Numérico

Sintaxe: GETGROUPS(**ref_campo**)

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve o parâmetro da função GETGROUPS.

Parâmetro	Descrição
ref_campo	Campo Lista de usuários/grupos ou Permissões de registro.

Exemplos:

A tabela a seguir fornece exemplos de fórmulas da função GETGROUPS.

Fórmula	Resultado
<p>IF(CONTAINS(ANY, GETGROUPS([Equipe]), GROUP(NAME, "Treinamento", "Suporte", "TI")), "Sim", "Não")</p> <p>em que o grupo "Treinamento" (neste exemplo, referido por nome, mas convertido em ID em tempo de execução) corresponde ao ID de um grupo retornado do campo Equipe, do tipo lista de usuários/grupos.</p>	Sim
<p>IF(ISEMPTY(GETGROUPS([Revisores]), "Em branco – Nenhum grupo selecionado", "Não está vazio – Um ou mais grupos selecionados")</p> <p>em que 1 ou mais grupos estão atualmente selecionados no campo Permissões de registro de revisores.</p> <p>Neste exemplo, ISEMPTY está avaliando apenas os valores de grupos e não de usuários selecionados no campo Revisores. Neste exemplo, ISEMPTY retornaria TRUE se nenhum grupo estivesse selecionado, mas 1 ou mais usuários estão selecionados.</p>	Não está vazio – 1 ou mais grupos selecionados
<p>IF(AND(ISEMPTY(GETUSERS ([Revisores]), ISEMPTY(GETGROUPS, ([Revisores])), "Em branco – Nenhum usuário ou grupo selecionado", "Não está vazio – 1 ou mais usuários ou grupos selecionados")</p> <p>em que não há valores selecionados no campo Permissões de registro de revisores.</p> <p>Para avaliar tanto usuários quanto grupos selecionados, 2 funções ISEMPTY podem ser incorporadas na função AND, como no exemplo abaixo.</p>	Em branco – Nenhum usuário ou grupo selecionado
<p>IF(AND(CONTAINS(EXACT, GETGROUPS([Gerentes de caso]), GROUP(NAME, "Treinamento", "Suporte", "TI")), CONTAINS(EXACT, GETUSERS([Gerentes de</p>	Todos os usuários e grupos de

Fórmula	Resultado
<p>caso]], USER(LOGIN, "blair.gates", "alfred.turks", "betty.smalls"))), "Todos os usuários e grupos de teste selecionados", "Teste com falha")</p> <p>em que o campo Permissões de registro de gerentes de caso contém todos os valores selecionados para teste e somente os valores selecionados.</p> <p>Para testar uma combinação de valores de usuário e grupo selecionados em um campo Lista de usuários/grupos ou Permissões de registro, 2 funções CONTAINS podem ser construídas (1 para testar grupos; 1 para testar usuários) e incorporadas à função AND, OR ou NOT.</p>	teste selecionados

Função GETUSERS

A função GETUSERS retorna uma lista de IDs de usuário dos usuários atualmente selecionados em um campo especificado do tipo Lista de usuários/grupos ou Permissões de registro. O ID de usuário deve ser avaliado por outra função. Por exemplo, a função GETUSERS pode ser usada dentro de uma função CONTAINS para avaliar se um determinado usuário corresponde ao ID de um usuário retornado de um campo especificado do tipo Lista de usuários/grupos.

Importante: Para garantir que a função retorne apenas 1 ID do usuário, configure a opção Máximo de seleções na seção Configuração para um campo Lista de usuários/grupos ou campo Permissões de registro como 1. Caso contrário, a função GETUSERS retornará um erro.

Tipo de retorno: Numérico

Sintaxe: GETUSERS(**ref_campo**)

Na sintaxe acima, o parâmetro em negrito é obrigatório.

A tabela a seguir descreve o parâmetro da função GETUSERS.

Parâmetro	Descrição
ref_campo	

Parâmetro	Descrição
	Campo Lista de usuários/grupos ou Permissões de registro.

Exemplos:

A tabela a seguir fornece exemplos de fórmulas da função GETUSERS.

Fórmula	Resultado
<p>IF(CONTAINS(ANY, GETUSERS ([Representante de vendas]), USER(NAME, "Wilson, Jonah", "Kellerman, Kathy", "Boone, Julia")), "Sim", "Não")</p> <p>em que a usuária "Boone, Julia" (neste exemplo, referida por nome, mas convertida em ID em tempo de execução) corresponde ao ID de um usuário retornado de um campo do tipo lista de usuários/grupos denominado Representante de vendas.</p>	Sim
<p>IF(ISEMPTY(GETUSERS([Colaborador]), "Em branco – Nenhum usuário selecionado", "Não está vazio – 1 ou mais usuários selecionados")</p> <p>em que nenhum usuário está atualmente selecionado no campo Lista de usuários/grupos colaboradores.</p> <p>Neste exemplo, ISEMPTY está avaliando apenas os valores de usuários e não de grupos selecionados no campo Colaborador. Neste exemplo, ISEMPTY retornaria TRUE se nenhum usuário estivesse selecionado, mas 1 ou mais grupos estão selecionados.</p>	Em branco – Nenhum usuário selecionado
<p>IF(AND(ISEMPTY(GETUSERS ([Revisores]), ISEMPTY(GETGROUPS, ([Revisores])), "Em branco – Nenhum usuário ou grupo selecionado", "Não está vazio – 1 ou mais usuários ou grupos selecionados")</p>	Em branco – Nenhum usuário ou grupo selecionado

Fórmula	Resultado
<p>em que não há valores selecionados no campo Permissões de registro de revisores.</p> <p>Para avaliar tanto usuários quanto grupos selecionados, 2 funções ISEMPTY podem ser incorporadas na função AND, como no exemplo abaixo.</p>	

Função GROUP

A função GROUP é usada para manter a validade da referência de uma fórmula a uma seleção de grupo específico em um campo Lista de usuários/grupos ou Permissões de registro, mesmo que o nome do grupo seja posteriormente alterado.

A função GROUP também aceita diretamente números de ID de grupo atribuídos pelo sistema. No sistema, cada grupo tem um número interno de ID que garante a identificação exclusiva. Por exemplo, se no sistema existirem 2 grupos denominados Suporte, o nome do grupo não poderá ser resolvido para determinar se o Suporte pretendido está selecionado no campo especificado, Lista de usuários/grupos ou Permissões de registro. No entanto, você pode usar o ID do sistema (por exemplo: 48761) em vez do nome ambíguo do grupo para identificar exatamente o grupo Suporte correto.

Tipo de retorno: Texto ou Numérico, dependendo do formato selecionado para o parâmetro tipo_ref

Sintaxe: GROUP(**tipo_rel**, **valor1**, valor2...)

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve os parâmetros da função GROUP.

Parâmetro	Descrição
tipo_ref	Aceita a palavra-chave NAME ou ID. Se for especificado NAME, a função avaliará os valores selecionados no campo por nome literal de grupo, por exemplo: "Suporte". Se for especificado ID, a função avaliará os

Parâmetro	Descrição
	valores de grupo selecionados com base nos números de ID exclusivos atribuídos pelo sistema.
valor1, valor2...	<p>Um ou mais valores em um campo Lista de usuários/grupos ou Permissões de registro.</p> <p>NOME. Se tipo_ref for NAME, os nomes de grupo devem ser especificados exatamente como exibidos no campo Lista de usuários/grupos ou Permissões de registro. A comparação diferencia maiúsculas e minúsculas. Se na função GROUP for usado NAME, ocorrerá um erro durante a validação da fórmula se nenhum destes argumentos for verdadeiro:</p> <ul style="list-style-type: none"> • Se o grupo nomeado não for encontrado em nenhum domínio • Se o grupo nomeado for encontrado em mais de 1 domínio • Se o grupo nomeado for encontrado em um só domínio, mas o nome do grupo não for exclusivo dentro daquele domínio <p>Se a string NAME contiver pelo menos 1 sinal de @, o sistema considerará todo o texto após o último sinal de @ uma referência ao domínio e tentará localizar esse domínio. Se a string NAME não contiver um sinal de @, o sistema procurará a correspondência exata à string inteira de nome do grupo no domínio do Archer (NULL) e no domínio padrão.</p> <p>Observação: O sistema tentará fazer a correspondência do nome do domínio com os domínios ativos e excluídos (por exemplo, aqueles excluídos do software). Somente nomes de domínio ativo devem ser exclusivos; é possível existir nome de domínio excluído igual ao de um domínio ativo. Se existir a string de nome de grupo em mais de 1 domínio com nome igual, ocorrerá falha no sistema na validação da fórmula.</p> <p>Se o domínio for encontrado:</p> <ol style="list-style-type: none"> 1. O sistema considerará todo o texto na string de nome antes do último sinal de @ como nome de

Parâmetro	Descrição
	<p>grupo e tentará encontrar esse grupo dentro do domínio.</p> <ol style="list-style-type: none"> 2. Se o grupo for encontrado dentro do domínio, o sistema substituirá a string de nome de grupo na fórmula pelo ID do grupo correspondente a aquele log-in. 3. Se o domínio não for encontrado: 4. O sistema procurará uma correspondência exata à toda a string de nome de grupo no domínio do Archer (NULL) e no domínio padrão. 5. Se existir apenas 1 grupo com esse nome, o sistema substituirá na fórmula a string de nome de grupo pelo ID do grupo. <p>ID. Se o parâmetro tipo_ref for ID, a função deverá retornar 1 ou mais números de ID de grupo atribuídos pelo sistema. Os IDs devem estar entre aspas na lista. Se na função GROUP for usado ID, ocorrerá um erro durante a validação da fórmula se o ID de grupo não for encontrado em nenhum domínio.</p>

Exemplos:

A tabela a seguir fornece exemplos de fórmulas da função GROUP.

Fórmula	Resultado
<p>IF(CONTAINS(ANY, GETGROUPS([Técnico]), GROUP(NAME, "Nível 1", "Nível 4", "Nível 9")), "Prioridade", "Padrão")</p> <p>cujo grupo "Nível 4" está selecionado no campo Lista de usuários/grupos Técnico.</p>	Prioridade
<p>IF(CONTAINS(ANY, GETGROUPS([Técnico]), GROUP(ID, 76712, 89766, 90287)), "Sim", "Não")</p> <p>cujo grupo "Nível 9" está selecionado no campo Lista de usuários/grupos Técnico e cujo ID de sistema exclusivo do grupo é 90287.</p>	Sim

Função ISCORRECT

A função ISCORRECT avalia uma pergunta do tipo Lista de valores e determina se o valor selecionado é identificado como "Correto" ou "Incorreto". Os valores são identificados como corretos ou incorretos na guia Resposta da página Definir campos. A função avalia como TRUE se o valor selecionado estiver definido como o valor "Correto". A função avalia como FALSE se o valor selecionado não estiver definido como o valor "Correto". A função ISCORRECT deve ser usada junto com a função IF e somente para avaliar uma pergunta do tipo Lista de valores dentro de um questionário.

Tipo de retorno: Texto, numérico, data ou seleção de campo Lista de valores, dependendo do tipo de dado fornecido para os parâmetros valor_se_verdadeiro e valor_se_falso.

Sintaxe: ISCORRECT([ref_campo])

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve o parâmetro da função ISCORRECT.

Parâmetro	Descrição
ref_campo	Uma referência à pergunta Lista de valores (por exemplo: [nome pergunta]).

Exemplo:

A tabela a seguir fornece um exemplo de fórmula da função ISCORRECT.

Fórmula	Resultado
IF(ISCORRECT([Pergunta sobre senha]),"Compatível", "Não compatível") cujo valor selecionado para a pergunta está identificado como "Correto".	Em conformidade

Função ISEEMPTY

A função ISEEMPTY é usada para determinar se um determinado campo contém um valor ou está em branco (vazio). A função avaliará como TRUE se o campo especificado estiver em branco (isto é, não contiver valor). A função avaliará como FALSE se o campo especificado não estiver em branco (isto é, contiver valor). A função ISEEMPTY deve ser usada junto com a função IF.

Tipo de retorno: Texto, numérico, data ou seleção de campo Lista de valores, dependendo do tipo de dado fornecido para os parâmetros valor_se_verdadeiro e valor_se_falso.

Sintaxe: ISEEMPTY([**ref_campo**])

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve o parâmetro da função ISEEMPTY.

Parâmetro	Descrição
ref_campo	Referência a um campo, por exemplo, [nome campo].

Exemplos:

A tabela a seguir fornece exemplos de fórmulas da função ISEEMPTY.

Fórmula	Resultado
IF(ISEEMPTY([Segundo nome]), "Sem segundo nome", "Segundo nome é " [Segundo nome]) em que Segundo nome é um campo Texto que não contém valor.	Sem segundo nome
IF(ISEEMPTY([Segundo nome]), "Sem segundo nome", "Segundo nome é " [Segundo nome]) em que Segundo nome é um campo Texto que contém o valor Douglas.	Segundo nome é Douglas

Função ISNUMBER

A função ISNUMBER verifica o valor especificado e retorna TRUE ou FALSE dependendo se for um número. Você pode usar esta função para obter informação sobre um valor antes de realizar um cálculo ou qualquer outra operação.

Tipo de retorno: TRUE ou FALSE

Sintaxe: ISNUMBER(**valor**)

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve o parâmetro da função ISNUMBER.

Parâmetro	Descrição
value	<p>Valor que você deseja testar. O argumento de valor pode estar em branco (célula vazia), ser um erro, valor lógico, texto, número ou valor de referência, ou um nome referente a qualquer um destes argumentos. Retorna TRUE se o valor se referir a um número.</p> <p>Observação: os argumentos de valor das funções IS não são convertidos. Qualquer valor numérico entre aspas duplas é considerado texto. Por exemplo, na maioria das outras funções em que um número é obrigatório, o valor de texto "19" é convertido no número 19. No entanto, na fórmula ISNUMBER("19"), "19" não é convertido de valor de texto em valor de número, e a função ISNUMBER retorna FALSE.</p>

Exemplo:

A tabela a seguir fornece um exemplo de fórmula da função ISNUMBER.

Fórmula	Resultado
ISNUMBER(4)	Verifica se 4 é um número (TRUE)

Função MOSTRECENTVALUE

A função MOSTRECENTVALUE exibe o valor específico de um registro, avaliado como o "mais recente" de uma lista de registros relacionados. Por exemplo, se um registro no aplicativo Instalações estiver relacionado a vários questionários, você pode usar esta função para retornar o valor do campo Resumo quantitativo do questionário enviado mais recentemente.

Tipo de retorno: Texto

Sintaxe: MOSTRECENTVALUE(**campo_para_exibir**, **campo_critérios_data**)

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve os parâmetros da função MOSTRECENTVALUE.

Parâmetro	Descrição
campo_para_exibir	Referência ao valor do campo que você deseja exibir (por exemplo, REF([Nome do campo Referência cruzada],[nome campo])).
campo_critérios_data	Campo Data usado para determinar o registro relacionado cujo valor seja o mais recente (por exemplo: REF([Nome do campo Referência cruzada],[Nome do campo Data])).

Exemplo:

A tabela a seguir fornece um exemplo de fórmula da função MOSTRECENTVALUE.

Fórmula	Resultado
<p>MOSTRECENTVALUE(REF([Questionário de risco],[Risco inerente]), REF([Questionário de risco],[Data de envio]))</p> <p>cujo valor no campo Pontuação intrínseca ao registro com o valor mais recente de Data de envio é "65".</p>	65

Função NOVALUE

A função NOVALUE é usada para definir um valor nulo para um campo calculado Data, Texto ou Numérico ou para definir um campo calculado Lista de valores para não ter seleção. A função NOVALUE é válida apenas no contexto da função IF.

Observação: a função NOVALUE não pode ser passada para a função VALUEOF para apagar seleções de um campo calculado Lista de valores.

Tipo de retorno: Nenhum

Sintaxe: NOVALUE()

Esta função não tem nenhum parâmetro.

Exemplos:

A tabela a seguir fornece um exemplo de fórmula da função NOVALUE.

Fórmula	Resultado
<p>IF([Classificação] >=0, "Ação necessária", NOVALUE())</p> <p>cujo valor de Classificação é menor que 6 e o campo calculado é um campo Texto.</p>	O campo calculado Texto está definido como nulo.

Função OTHERTEXT

A função OTHERTEXT retorna o texto digitado por um usuário no campo "Outros" para o campo do tipo Lista de valores ou a pergunta do tipo Lista de valores especificados.

Tipo de retorno: Texto

Sintaxe: OTHERTEXT([**ref_campo**])

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve o parâmetro da função OTHERTEXT.

Parâmetro	Descrição
ref_campo	Referência a um campo Lista de valores ou a uma pergunta Lista de valores (por exemplo: [nome pergunta]).

Exemplos:

A tabela a seguir fornece um exemplo de fórmula da função OTHERTEXT.

Fórmula	Resultado
OTHERTEXT([Severidade do último incidente de segurança]) cujo nome da pergunta do tipo Lista de valores é "Severidade do último incidente de segurança" e o texto digitado no campo Outros é "Não tivemos incidente de segurança".	Não tivemos incidente de segurança.

Função REF

A função REF retorna uma referência a um campo que seja campo filho de um campo pai Subformulário, Referência cruzada, Registros relacionados ou Agendador . Ela pode ser usada em outro cálculo que necessite de

informações de um conjunto de campos ou valores. A função REF é válida para ser usada apenas com estes tipos de campo:

- Referência cruzada
- Registros relacionados
- Subformulário
- Agendador

Tipo de retorno: Texto, numérico, data ou seleção de campo Lista de valores, dependendo do tipo de dado retornado do campo referido

Sintaxe: REF(campo_pai, campo_filho, nome_nível_dados)

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve os parâmetros da função REF.

Parâmetro	Descrição
campo_pai	O nome do campo Referência cruzada, Registros relacionados, Subformulário ou Agendador. Esses tipos de campo são considerados "pai" porque atuam como contêineres de outros campos. Por exemplo, um campo Subformulário em si não tem nenhum valor; os valores só podem ser derivados de seus campos filho.
campo_filho	Nome de um campo contido em um campo_pai.
nome_nível_dados	<p>Se um campo Referência cruzada referir-se a um aplicativo em vários níveis, nome do nível em que reside o campo. É opcional a passagem do parâmetro nome_nível_dados e necessária somente quando um campo Referência cruzada refere-se a um aplicativo em vários níveis.</p> <p>Para um campo Agendador, nome_nível_dados é obrigatório, já que os campos Agendador sempre se referem a vários níveis. O campo agendador sempre se refere ao aplicativo Compromisso e pode referir-se ao aplicativo de recursos ou a 1 ou mais aplicativos pais.</p>

Exemplos:

A tabela a seguir fornece exemplos de fórmulas da função REF.

Fórmula	Resultado
<p>SUM(REF([Controles], [Risco]))</p> <p>em que Controles é um campo Referência cruzada e Risco é um campo no aplicativo simples com a referência cruzada.</p>	Referência cruzada com o aplicativo simples
<p>AVERAGE(REF([Medidas de resposta], [Classificação de severidade], [Respostas]))</p> <p>em que Medidas de resposta é um campo Referência cruzada, Respostas é um nível do aplicativo de vários níveis com referência cruzada e Classificação de severidade é um campo no nível de dados Respostas.</p>	Referência cruzada ao aplicativo em vários níveis
<p>COUNTA(REF([Linhas de base], [Nome]))</p> <p>em que Linhas de base é um campo Registros relacionados e Nome é um campo do aplicativo que contém o campo Referência cruzada correspondente.</p>	Registros relacionados
<p>COUNTA(REF([Ações], [ID do contato]))</p> <p>em que Ações é um campo Subformulário e ID do contato é um campo no subformulário associado ao campo Subformulário.</p>	Subformulário
<p>SUM(REF([CampoAgendador],[Duração (horas)], [Compromisso]))</p> <p>em que CampoAgendador é um campo do agendador, Duração (horas) é um campo no aplicativo Compromisso e Compromisso é o nível</p>	Referência do campo Agendador ao aplicativo Compromisso

Fórmula	Resultado
por meio do qual o agendador está fazendo referência.	

Função **SELECTEDVALUENUMBER**

A função **SELECTEDVALUENUMBER** extrai o valor numérico do item de lista de valores selecionado de um campo do tipo Lista de valores. Se o campo Lista de valores permitir várias seleções, esta função deverá ser usada juntamente com uma função agregada, como mostrado no exemplo abaixo:

SUM(SELECTEDVALUENUMBER([Campo de seleção múltipla Lista de valores]))

Além disso, se você fizer referência a um campo Lista de valores em um aplicativo com referência cruzada, tanto o campo Referência cruzada usado para formar a relação com o aplicativo quanto o campo Lista de valores do aplicativo relacionado devem ser campos de seleção única para que possa ser usada a função **SELECTEDVALUENUMBER** sem incorporá-la a uma função agregada. Se o campo do tipo Referência cruzada ou Lista de valores permitir seleções múltiplas, será necessário usar uma função agregada, como no exemplo abaixo:

AVERAGE(SELECTEDVALUENUMBER(REF([Campo de seleção múltipla Referência cruzada], [Campo Lista de valores]))))

Se nenhum valor numérico for atribuído a um item da lista de valores, esse valor será considerado 0. A única exceção é quando a função **AVERAGE** é usada juntamente com a função **SELECTEDVALUENUMBER**. Nesse caso, o valor nulo não será usado no cálculo. Use como exemplo os seguintes valores:

Valor A: 10

Valor B: 5

Valor C: nenhum valor numérico atribuído

Se a função **SUM** for usada junto com a função **SELECTEDVALUENUMBER**, o Valor C será considerado 0 no cálculo. Se os 3 valores estivessem selecionados no campo Lista de valores, o resultado do cálculo seria 15. No entanto, se a função **AVERAGE** fosse usada e os 3 valores estivessem

selecionados no campo Lista de valores, Valor C seria ignorado no cálculo porque não teria valor numérico. O resultado seria 7,5.

Tipo de retorno: Numérico

Sintaxe: SELECTEDVALUENUMBER(**ref_campo**)

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve o parâmetro da função SELECTEDVALUENUMBER.

Parâmetro	Descrição
ref_campo	Referência a um campo Lista de valores no aplicativo (por exemplo: [Classificação de risco]), um campo em um subformulário filho (por exemplo: [Notas].[Classificação de risco]) ou um campo em um aplicativo com referência cruzada (por exemplo: [Fornecedores][Classificação de risco]).

Exemplos:

A tabela a seguir fornece um exemplo de fórmula da função SELECTEDVALUENUMBER.

Fórmula	Resultado
SELECTEDVALUENUMBER([Classificação de risco]) cujo valor selecionado no campo Classificação de risco é "Alto" e o valor numérico atribuído ao valor "Alto" é 10.	10
MAX(SELECTEDVALUENUMBER([Departamentos afetados])) em que Departamentos afetados é um campo de seleção múltipla Lista de valores, os valores selecionados são "Operações" e "TI" e os valores numéricos associados são 7 e 10, respectivamente.	10

Fórmula	Resultado
<p>AVERAGE(SELECTEDVALUENUMBER(REF[Fornecedores], [Classificação de risco]))</p> <p>em que Fornecedores é um campo de seleção múltipla Referência cruzada ao aplicativo Fornecedores, Classificação de risco é um campo de seleção única Lista de valores no aplicativo Fornecedores, os valores selecionados nos registros relacionados são "Alto" e "Baixo", e os valores numéricos associados são 10 e 8, respectivamente.</p>	9

Função TRACKINGID

A função TRACKINGID retorna um ID de registro que identifica exclusivamente o registro atual em todos os aplicativos. Esta função pode ser usada junto com a função CONTENTID para produzir um ID complexo combinando o ID de todo o sistema, o ID específico do aplicativo e os dados extraídos de outros campos.

Tipo de retorno: Numérico

Sintaxe: TRACKINGID()

Esta função não tem parâmetros.

Exemplo:

A tabela a seguir fornece uma fórmula de exemplo da função TRACKINGID.

Fórmula	Resultado
<p>[Código do tipo] & "-" & TRACKINGID() & "-" & CONTENTID()</p> <p>cujo valor no campo Código do tipo é WORM, o ID de controle de todo o sistema é 678904 e o ID de registro de conteúdo específico ao aplicativo é 34.</p>	WORM-678904-34

Função USER

A função USER mantém a validade da referência de uma fórmula a uma seleção de grupo específico em um campo Lista de usuários/grupos ou Permissões de registro, mesmo que o nome do grupo seja posteriormente alterado. As referências ao usuário podem ser passadas como nomes literais ou IDs de log-on. Opcionalmente um domínio específico pode fazer referência a um nome de log-on.

Por exemplo, se o nome de usuário literal "Jones, Mary" for referido em uma fórmula dentro da função USER e o nome de usuário for posteriormente alterado para "Jones-Smith, Mary", a referência original ao nome de usuário será automaticamente atualizada na fórmula para "Jones-Smith, Mary".

Outro exemplo, a usuária Mary Jones é mencionada em uma fórmula com o nome de log-on no Archer para o domínio "bigcompany.com". O nome de log-on da usuária deve ser passado para a função USER como "mjones@bigcompany.com". Se um administrador posteriormente alterar para "msmith" o nome de log-on de Mary Jones naquele domínio, o nome de log-on original será automaticamente atualizado para "msmith@bigcompany.com".

A função USER também aceita diretamente números de ID do usuário atribuídos pelo sistema. No sistema, cada usuário tem um número interno de ID que garante a identificação exclusiva. Por exemplo, se no sistema existirem 2 usuários chamados Graham, Ned, o nome de usuário não poderá ser avaliado para determinar se o Graham, Ned pretendido está selecionado no campo especificado Lista de usuários/grupos ou Permissões de registro. No entanto, você pode usar o ID do sistema, por exemplo, 76219 no lugar do nome ambíguo de usuário para identificar exclusivamente o Graham, Ned correto.

Tipo de retorno: Texto ou Numérico, dependendo do formato selecionado para o parâmetro tipo_ref

Sintaxe: USER(**tipo_ref**, **valor1**, valor2...)

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve os parâmetros da função USER.

Parâmetro	Descrição
tipo_ref	

Parâmetro	Descrição
	<p>Aceita a palavra-chave NAME, LOGIN ou ID.</p> <ul style="list-style-type: none"> • Se for especificado NAME, a função avaliará os valores selecionados no campo por nome literal de usuário, por exemplo, "Jones, Mary". • Se for especificado LOGIN, a função avaliará os valores de usuário selecionados com base no nome de log-on, e não no nome de usuário, por exemplo, mjones@bigcompany.com. • Se for especificado ID, a função avaliará os valores de usuário selecionados com base nos números de ID exclusivos atribuídos pelo sistema.
valor1, valor2...	<p>Um ou mais valores em um campo Lista de usuários/grupos ou Permissões de registro.</p> <ul style="list-style-type: none"> • NOME. Se o parâmetro tipo_ref for NAME, os nomes de usuário deverão ser especificados exatamente como exibidos no campo Lista de usuários/grupos ou Permissões de registro. O nome de usuário deve ser especificado no seguinte formato: "sobrenome, nome". A comparação diferencia maiúsculas e minúsculas. O sistema testará apenas usuários não excluídos. • Se na função USER for usado NAME, ocorrerá um erro durante a validação da fórmula se nenhum destes argumentos for verdadeiro: <ul style="list-style-type: none"> ◦ Se o usuário nomeado não for encontrado em nenhum domínio ◦ Se o usuário nomeado for encontrado em mais de 1 domínio ◦ Se o usuário nomeado for encontrado em um só domínio, mas o nome de usuário não for exclusivo dentro daquele domínio. • LOGIN. Se o parâmetro tipo_ref for LOGIN, a função deverá retornar 1 ou mais valores de nome de log-on de usuário do Archer. A verificação de correspondência será feita com base nos log-ons de usuário e não nos nomes e sobrenomes.

Parâmetro	Descrição
	<ul style="list-style-type: none"> • Se na função USER for usado LOGIN, ocorrerá um erro durante a validação da fórmula se qualquer um destes argumentos for verdadeiro: <ul style="list-style-type: none"> ◦ Se o usuário mencionado pelo logon não for encontrado em nenhum domínio ◦ Se o usuário mencionado pelo logon for encontrado em mais de 1 domínio <p>Se a string LOGIN contiver pelo menos 1 sinal de @, o sistema considerará todo o texto após o último sinal de @ uma referência ao domínio e tentará localizar esse domínio. Se a string LOGIN não contiver um sinal de @, o sistema procurará fazer a correspondência exata à string de log-in inteira de nome do grupo no domínio do Archer (NULL) e no domínio padrão.</p> <p>Observação: O sistema tentará fazer a correspondência do nome do domínio com os domínios ativos e excluídos, por exemplo, aqueles excluídos do software. Somente nomes de domínio ativo devem ser exclusivos; é possível existir nome de domínio excluído igual ao de um domínio ativo. Se existir a string de log-on em mais de 1 domínio com nome igual, ocorrerá falha no sistema na validação da fórmula.</p> <ul style="list-style-type: none"> • Se o domínio for encontrado: <ol style="list-style-type: none"> 1. O sistema considerará todo o texto na string de logon antes do último sinal @ como o nome de logon de usuário e tentará encontrá-lo no domínio. 2. Se o logon for encontrado no domínio, o sistema substituirá a string de logon da fórmula pelo ID do usuário correspondente ao logon. <p>Se o domínio não for encontrado:</p> <ol style="list-style-type: none"> 1. O sistema procurará uma correspondência exata com toda a string de logon no domínio do Archer (NULL) e no domínio padrão.

Parâmetro	Descrição
	<p>2. Se existir apenas 1 usuário com o login em questão, o sistema substituirá a string de login da fórmula pelo ID do usuário correspondente ao login.</p> <ul style="list-style-type: none"> • ID. Se o parâmetro tipo_ref for ID, a função deverá retornar 1 ou mais números de ID de usuário atribuídos pelo sistema. IDs de usuário são atribuídos pelo sistema e são sempre exclusivos. IDs de usuário são números e não devem ser digitados entre aspas. Se na função USER for usado o ID, ocorrerá um erro durante a validação da fórmula se um dos argumentos for verdadeiro e o ID de usuário referido não for encontrado em nenhum domínio.

Exemplos:

A tabela a seguir fornece exemplos de fórmulas da função USER.

Fórmula	Resultado
<p>IF(CONTAINS(ANY, GETUSERS([Técnico]), USER(NAME, "Thurman, Laurie", "Winters, George")), "Padrão", "Prioridade")</p> <p>em que a usuária "Thurman, Laurie" está selecionada no campo Lista de usuários/grupos técnico.</p>	Padrão
<p>IF(CONTAINS(ANY, GETUSERS([Revisor]), USER(NAME, "Jasper, Susan", "Miner, Burt", "Rollins, Jacob")), "Sim", "Não")</p> <p>em que o usuário "Miner, Burt" está selecionado no campo Revisor de permissões de registro.</p>	Sim
<p>IF(CONTAINS(ANY, GETUSERS([Gerente]), USER(LOGIN, "kjackson", "tbarnett@bigcompany.com",</p>	O valor "Escalar" está selecionado

Fórmula	Resultado
<p>"jwilson@bigcompany.net", "smartin")),VALUEOF([Alerta], "Escalar"), VALUEOF([Alerta], NOVALUE())</p> <p>em que "Barnett, Tina" está selecionada no campo Lista de usuários/grupos e o ID de log-on no Archer para o domínio "bigcompany.com" é "tbarnett".</p>	no campo Lista de valores de alerta.
<p>IF(CONTAINS(ANY, GETUSERS([Colaboradores]), USER(ID, 76299, 56897, 79867)), "Localizado", "Não localizado")</p> <p>em que "Eastman, Tina" está selecionada no campo Lista de usuários/grupos e o ID de usuário no sistema é 79867.</p>	Localizado

Função USERDEFAULTEMAIL

A função USERDEFAULTEMAIL retorna o endereço de e-mail do criador ou do editor do registro. Criador é o usuário que criou o registro (ou que está atualmente criando o registro). Editor é o usuário que editou por último o registro.

Tipo de retorno: Texto

Sintaxe: USERDEFAULTEMAIL(**tipo_de_usuario**)

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve o parâmetro da função USERDEFAULTEMAIL.

Parâmetro	Descrição
tipo_usuario	Aceita a palavra-chave CREATOR ou EDITOR. Se CREATOR for especificado, a função retornará o endereço de e-mail padrão que o usuário associou à criação do registro. Se EDITOR for fornecido, a função retornará o endereço de e-mail padrão do usuário associado à atualização de registro mais recente.

Exemplos:

A tabela a seguir fornece exemplos de fórmulas da função USERDEFAULTEMAIL.

Fórmula	Resultado
"Olá, " & USERDEFAULTEMAIL(CRIADOR) Onde Janet@archer.com é o endereço de e-mail padrão do criador do registro.	Olá, Janet@archer.com
"Olá, " & USERDEFAULTEMAIL(EDITOR) Em que Miles@archer.com é o endereço de e-mail padrão do editor do registro.	Olá, Miles@archer.com

Função USERFIRSTNAME

A função USERFIRSTNAME retorna o nome do criador ou do editor do registro. Criador é o usuário que criou o registro (ou que está atualmente criando o registro). Editor é o usuário que editou por último o registro.

Tipo de retorno: Texto

Sintaxe: USERFIRSTNAME(**tipo_usuario**)

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve o parâmetro USERFIRSTNAME.

Parâmetro	Descrição
tipo_usuario	Aceita a palavra-chave CREATOR ou EDITOR. Se for especificado CREATOR, a função retornará o nome do usuário associado à criação do registro. Se for especificado EDITOR, a função retornará o nome do usuário associado à última atualização do registro.

Exemplos:

A tabela a seguir fornece exemplos de fórmulas da função USERFIRSTNAME.

Fórmula	Resultado
"Olá, " & USERFIRSTNAME(CREATOR) cujo nome do criador do registro é Janet.	Olá, Janet
"Olá, " & USERFIRSTNAME(EDITOR) cujo nome do editor do registro é Miles.	Olá, Miles

Função USERLASTNAME

A função USERLASTNAME retorna o sobrenome do criador ou do editor do registro. Criador é o usuário que criou o registro (ou que está atualmente criando o registro). Editor é o usuário que editou por último o registro.

Tipo de retorno: Texto

Sintaxe: USERLASTNAME(**tipo_usuario**)

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve o parâmetro USERLASTNAME.

Parâmetro	Descrição
tipo_usuario	Aceita a palavra-chave CREATOR ou EDITOR. Se for especificado CREATOR, a função retornará o sobrenome do usuário associado à criação do registro. Se for especificado EDITOR, a função retornará o sobrenome do usuário associado à última atualização do registro.

Exemplos:

A tabela a seguir fornece exemplos de fórmulas da função USERLASTNAME.

Fórmula	Resultado
"Sobrenome: " & USERLASTNAME(CREATOR) em que Rossi é o sobrenome do criador do registro.	Sobrenome: Rossi
"Sobrenome: " & USERLASTNAME(EDITOR) em que Eldrich é o sobrenome do editor do registro.	Sobrenome: Eldrich

Função USERMIDDLENAME

A função USERMIDDLENAME retorna o segundo nome do criador ou do editor do registro. Criador é o usuário que criou o registro (ou que está atualmente criando o registro). Editor é o usuário que editou por último o registro.

Tipo de retorno: Texto

Sintaxe: USERMIDDLENAME(**tipo_usuario**)

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve o parâmetro USERMIDDLENAME.

Parâmetro	Descrição
tipo_usuario	Aceita a palavra-chave CREATOR ou EDITOR. Se for especificado CREATOR, a função retornará o segundo nome do usuário associado à criação do registro. Se for especificado EDITOR, a função retornará o segundo nome do usuário associado à última atualização do registro.

Exemplos:

A tabela a seguir fornece exemplos de fórmulas da função USERMIDDLENAME.

Fórmula	Resultado
"Segundo nome: " & USERMIDDLENAME(CREATOR) em que Ellen é o segundo nome do criador do registro.	Segundo nome: Ellen
"Segundo nome: " & USERMIDDLENAME(EDITOR) em que Quentin é o segundo nome do criador do registro.	Segundo nome: Quentin

Função VALUEOF

A função VALUEOF mantém a validade da seleção de um campo do tipo Lista de valores. Se o texto de um valor na lista global ou personalizada de valores for alterado pelo administrador Archer, essa função automaticamente atualizará a fórmula para utilizar o novo texto para o valor. Por exemplo, se o valor "Azul" for referido em uma fórmula na função VALUEOF e depois for alterado para "Vermelho" na lista de valores, a referência do valor "Azul" será automaticamente alterada para "Vermelho" na fórmula.

Da perspectiva de um usuário, a função VALUEOF serve a 2 fins adicionais, ela possibilita que você:

- Avalie a presença de um valor específico em um campo Lista de valores.
- Defina seleções de valor em um campo Lista de valores.

O exemplo a seguir ilustra como utilizar a função VALUEOF em uma fórmula para um campo calculado Lista de valores. A função VALUEOF é utilizada para os parâmetros "valor_se_verdadeiro" e "valor_se_falso" na sintaxe da função IF.

```
IF([Classificação de risco]=10, VALUEOF([Importância], "Alto"),  
VALUEOF([Importância], "Baixo"))
```

Em um registro, essa fórmula avaliará a Classificação de risco e, se o valor nesse campo for maior que ou igual a 10, a fórmula selecionará o valor Alto no campo Lista de valores calculados por importância. Se o valor em Classificação de risco for menor que 10, a fórmula selecionará o valor Baixo no campo Importância.

Este segundo exemplo ilustra como utilizar a função VALUEOF em uma fórmula para um campo calculado Texto. A função VALUEOF é utilizada para o parâmetro valor na sintaxe da função CONTAINS.

```
IF(CONTAINS(ANY, [Local],VALUEOF([Local], "Nova  
Iorque")),"Sim","Não")
```

Em um registro, essa fórmula produzirá o valor "Sim" ou "Não" no campo Texto calculado caso o valor "Nova Iorque" tenha sido selecionado no campo Local, que é um campo do tipo Lista de valores. Se o valor "Nova Iorque" estiver selecionado, o campo Texto calculado exibirá o valor "Sim". Se o valor "Nova Iorque" não estiver selecionado, o campo Texto calculado exibirá o valor "Não".

Tipo de retorno: Texto

Sintaxe: VALUEOF(**campo_ref**, **valor1**, valor2...)

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve os parâmetros da função VALUEOF.

Parâmetro	Descrição
ref_campo	Referência a um campo, por exemplo, [nome campo].

Parâmetro	Descrição
valor1, valor2, . . .	<p>Valor em um campo Lista de valores. Informe o valor como " valor", por exemplo ,"Urgente ". Se o campo Lista de valores permitir várias seleções, vários valores poderão ser informados como " valor1", "valor2" e "valor3".</p> <p>VALUEOF([Values List Name], "value1", "value2", "value3"). Por exemplo, VALUEOF ([estados], "Washington", " Nova Iorque ", "Massachusetts")</p>

Exemplos:

A tabela a seguir fornece exemplos de fórmulas da função VALUEOF.

Fórmula	Resultado
<p>IF(CONTAINS(ANY, [Local],VALUEOF([Escritório], "Chicago", "Local", "Global")))</p> <p>cujo valor selecionado em Local é Chicago.</p>	Local
<p>IF(CONTAINS(ANY, [Região], VALUEOF([Região], "Nordeste")), VALUEOF([Escritório], "Connecticut", "Nova Iorque", "Massachusetts"), VALUEOF([Escritório], "Kansas", "Illinois", "Texas"))</p> <p>cujo valor selecionado em Região é Nordeste.</p>	<p>Connecticut</p> <p>New York</p> <p>Massachusetts</p>
<p>IF(CONTAINS(ANY, [Cor], VALUEOF([Cor], "Vermelho", "Verde", "Azul", "Amarelo")), "Cor primária", "Outros")</p> <p>cujo valor selecionado no campo Cor é Prata.</p>	Outro

Função WEIGHTEDSCORE

A função WEIGHTEDSCORE retorna o valor de pontuação ponderada dos valores selecionados em uma pergunta do tipo Lista de valores. O uso desta função é recomendado apenas se você atribuiu um peso à pergunta Lista de valores e um valor numérico a cada resposta possível.

Esta função pode ser usada para executar os seguintes cálculos:

$[valor_numérico_seleção_lista_valores] * [peso_pergunta_lista_valores]$
= WEIGHTEDSCORE

ou (para perguntas de seleção múltipla de Lista de valores):

$SUM([valor1_numérico_seleção_lista_valores],$
 $[valor2_numérico_seleção_lista_valores]) * [peso_pergunta_lista_valores]$
= WEIGHTEDSCORE

A função WEIGHTEDSCORE só pode ser usada em um questionário e referenciar apenas uma pergunta do tipo Lista de valores.

Tipo de retorno: Numérico

Sintaxe: **WEIGHTEDSCORE**(ref_campo)

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve o parâmetro da função WEIGHTEDSCORE.

Parâmetro	Descrição
ref_campo	Uma referência à pergunta Lista de valores (por exemplo: [nome pergunta]).

Exemplo:

A tabela a seguir fornece um exemplo de fórmula da função WEIGHTEDSCORE.

Fórmula	Resultado
WEIGHTEDSCORE([Acesso de novo usuário])	50

Fórmula	Resultado
cujo valor do peso da pergunta Acesso de novo usuário é "10" e o valor numérico da resposta é "5".	

Função **WEIGHTING**

A função **WEIGHTING** retorna o valor ponderado de uma pergunta do tipo Lista de valores. A função **WEIGHTING** só pode ser usada em um questionário e referenciar apenas uma pergunta do tipo Lista de valores.

Tipo de retorno: Numérico

Sintaxe: **WEIGHTING**(ref_campo)

Na sintaxe acima, os parâmetros em negrito são obrigatórios.

A tabela a seguir descreve o parâmetro da função **WEIGHTING**.

Parâmetro	Descrição
ref_campo	Uma referência à pergunta Lista de valores (por exemplo: [nome pergunta]).

Exemplo:

A tabela a seguir fornece um exemplo de fórmula da função **WEIGHTING**.

Fórmula	Resultado
WEIGHTING([Acesso de novo usuário]) cujo valor do peso da pergunta Acesso de novo usuário é "10"	10